# Maintainer: Jaroslav Lichtblau <svetlemodry@archlinux.org>
# Contributor: Ray Rashif <schiv@archlinux.org>
# Contributor: Andrea Scarpino <andrea@archlinux.org>
# Contributor: neuromante <lorenzo.nizzi.grifi@gmail.com>

pkgname=skrooge
pkgver=2.31.0
pkgrel=1
pkgdesc="A personal finances manager for KDE"
arch=('loong64' 'x86_64')
url="https://skrooge.org/"
license=('GPL')
depends=('qt5-base' 'qca-qt5' 'qt5-script' 'qt5-webengine' 'qt5-xmlpatterns' 'libofx'
         'grantlee' 'knewstuff5' 'krunner5' 'knotifyconfig5' 'kparts5'
         'sqlcipher' 'python' 'hicolor-icon-theme')
makedepends=('cmake' 'kdoctools5' 'extra-cmake-modules' 'qt5-tools' 'kdesignerplugin')
options=('!emptydirs')
changelog=$pkgname.changelog
source=(https://download.kde.org/stable/$pkgname/$pkgname-$pkgver.tar.xz{,.sig})
validpgpkeys=('103DC6D087B3A4A2ED975D7EF0C8CED3382A5C4D') #Stephane MANKOWSKI (Perso) <stephane@mankowski.fr>
sha256sums=('4bdd2c50a26453080f006948bb2374e5ae43a1e85316914e9d52cc17c01cf872'
            'SKIP')

build() {
  cd $pkgname-$pkgver

  cmake . -DSKG_WEBENGINE=ON \
          -DSKG_BUILD_TEST=OFF
  make
}

package() {
  cd $pkgname-$pkgver

  make DESTDIR="${pkgdir}" install
}
